

# Visualizing Multilevel Models

Andy Grogan-Kaylor

2024-05-07

## Table of contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>Organizing Questions</b>	<b>2</b>
<b>3</b>	<b>Setup</b>	<b>3</b>
<b>4</b>	<b>Get Data</b>	<b>3</b>
<b>5</b>	<b>Scatterplots (<code>twoway scatter y x</code>)</b>	<b>3</b>
<b>6</b>	<b>Simple Linear Fit (<code>twoway lfit y x</code>)</b>	<b>4</b>
<b>7</b>	<b>Linear Fit With Confidence Interval (<code>twoway lfitci y x</code>)</b>	<b>5</b>
<b>8</b>	<b>Combine Scatterplot and Linear Fit (<code>twoway (scatter y x) (lfit y x)</code>)</b>	<b>6</b>
<b>9</b>	<b>Spaghetti Plots (<code>spagplot y x, id(group)</code>)</b>	<b>7</b>
<b>10</b>	<b>Small Multiples (<code>twoway y x, by(group)</code>)</b>	<b>8</b>
<b>11</b>	<b>Taking A Random Sample</b>	<b>9</b>
<b>12</b>	<b>Multivariate (Predicted) Relationships</b>	<b>10</b>
12.1	Using Predicted Values ( <code>predict</code> ) . . . . .	11
12.1.1	Estimate The Model . . . . .	11
12.1.2	Generate Predicted Values . . . . .	12
12.1.3	Graph With <code>twoway</code> Syntax . . . . .	12
12.1.4	Spaghetti Plot With Predicted Values . . . . .	14
12.2	<code>margins</code> and <code>marginsplot</code> . . . . .	14
12.2.1	Estimate The Model . . . . .	14

12.2.2	Generate Predicted Values <i>At Specified Values</i> With <code>margins</code> . . . . .	15
12.2.3	Graph With <code>marginsplot</code> . . . . .	16
<b>13</b>	<b>Scatterplot With Linear Fit and Marginal Density Plots (<code>twoway</code> ...)</b>	<b>17</b>
13.1	Manually Generate The Densities To Plot Them Below ( <code>kdensity</code> ...)	17
13.2	Rescale The Densities So They Plot Well . . . . .	18
13.3	Make The Graph ( <code>twoway</code> ...)	18
13.4	Spaghetti Plot With Linear Fit and Marginal Density Plots . . . . .	19
<b>14</b>	<b>Curvilinear and Linear Fits</b>	<b>19</b>
	<b>Random Effects</b>	<b>19</b>

## 1 Introduction

An evolving set of notes on visualizing results from multilevel models.

The examples below use the `simulated_multilevel_data.dta` file from my draft text book on *Multilevel Thinking*. Here is a [direct link](#) to download the data.

This document relies on the extraordinary `Statamarkdown` library (Hemken 2023).

## 2 Organizing Questions

Try to think about some of the advantages and disadvantages of different approaches to visualizing multilevel models. In multilevel models, we don't want to just *control for* variation, but to start to *explore* the variation. Put concretely:

- Some approaches use *dots*. Some approaches use *lines*. Some approaches use *dots and lines*.
- Some approaches use the *raw unadjusted* data. Other approaches use *adjusted or model predicted* data.
- Some approaches attempt to show the *Level 2 specific regression lines*; some approaches only show an *average regression line*.
- What approaches might work well with *large numbers* of Level 2 units? What approaches might work well with *smaller numbers* of Level 2 units?

What approach(es) do you prefer?

### 3 Setup

I am not terrifically fond of the default `s2color` graph scheme in earlier versions of Stata. Here I make use of the `michigan` graph scheme available at: <https://agrogan1.github.io/Stata/michigan-graph-scheme/>.

```
set scheme michigan
```

Stata's `s1color` scheme—available in newer versions of Stata—would also be an option as would be Asjad Naqvi's incredible `schemepack`: <https://github.com/asjadnaqvi/stata-schemepack>.

Throughout the tutorial, I make frequent use of the `mcolor(%30)` option to add some visual interest to scatterplots by adding transparency to the markers.

### 4 Get Data

```
use "https://github.com/agrogan1/multilevel-thinking/raw/main/simulate-and-analyze-multilevel"
```

### 5 Scatterplots (twoway scatter y x)

```
twoway scatter outcome warmth, mcolor(%30)  
graph export myscatter.png, width(1500) replace
```

file myscatter.png saved as PNG format




Figure 1: Scatterplot

## 6 Simple Linear Fit (twoway lfit y x)

```
twoway lfit outcome warmth
graph export mylinear.png, width(1500) replace
```

file mylinear.png saved as PNG format




Figure 2: Linear Fit

## 7 Linear Fit With Confidence Interval (`twoway lfitci y x`)

```
twoway lfitci outcome warmth  
graph export mylfitci.png, width(1500) replace
```

file mylfitci.png saved as PNG format




Figure 3: Linear Fit With Confidence Interval

## 8 Combine Scatterplot and Linear Fit (twoway (scatter y x) (lfit y x))

```
twoway (scatter outcome warmth, mcolor(%30)) (lfit outcome warmth)
graph export myscatterlinear.png, width(1500) replace
```

file myscatterlinear.png saved as PNG format




Figure 4: Scatterplot and Linear Fit

## 9 Spaghetti Plots (spagplot y x, id(group))

```
spagplot outcome warmth, id(country)
graph export myspaghetti.png, width(1500) replace
```

file myspaghetti.png saved as PNG format




Figure 5: Spaghetti Plot

## 10 Small Multiples (twoway y x, by(group))


*Small Multiples*, showing a separate graph for each group in the data, are an increasingly popular data visualization technique. Below, I build a small multiples graph using the `by` option in Stata. I use the `aspect` option to adjust the *aspect ratio* of the graph for better visual presentation.

```
twoway (scatter outcome warmth, mcolor(%30)) ///
(lfit outcome warmth), ///
by(country) aspect(1)

graph export mysmallmultiples.png, width(1500) replace
```

file mysmallmultiples.png saved as PNG format





Graphs by country id

Figure 6: Small Multiples

## 11 Taking A Random Sample

At times, we may have *too many* Level 2 units to effectively display them on a *spaghetti plot*, or using *small multiples*. If this is the case, we may need to *randomly sample* Level 2 units. This can be difficult to accomplish as our standard `sample` command operates on each row, or on Level 1 units.

We can accomplish random sampling at Level 2, with a little bit of code.

```
set seed 3846 // random seed for reproducibility

gen randomid = runiform() // generate a random id variable

* by country (i.e. by Level 2 unit) replace the randomid
* with the first randomid for that country (Level 2 unit)
* so that every person in that country has the same random id

bysort country: replace randomid = randomid[1]

summarize randomid // descriptive statistics for random id

twoway (scatter outcome warmth, mcolor(%30)) /// scatterplot
```

```
(lfit outcome warmth) /// linear fit
if randomid < .5, /// only use a subset of randomids
by(country) aspect(1) // by country

graph export mysmallmultiples2.png, width(1500) replace
```

(2,970 real changes made)

Variable	Obs	Mean	Std. dev.	Min	Max
randomid	3,000	.6174022	.2374704	.0733026	.9657055

file mysmallmultiples2.png saved as PNG format




Figure 7: Small Multiples With A Random Sample Of Countries

## 12 Multivariate (Predicted) Relationships

A sometimes unacknowledged point is that graphs—unless we take steps to correct this—reflect *unadjusted*, or *bivariate* associations. We may sometimes wish to develop a graphs that reflect the *adjusted* or *predicted* estimates from our models.

## 12.1 Using Predicted Values (predict)

`predict` generates a predicted value for *every observation in the data*.

In multilevel models, *prediction* is a complex question. Prediction may—or may not—incorporate the information from the random effects. The procedures below outline graphs that incorporate predictions using the random effects, by using the `predict ...`, fitted syntax.

### 12.1.1 Estimate The Model

```
mixed outcome warmth physical_punishment i.intervention || country: // estimate MLM
```

Performing EM optimization Performing gradient-based optimization:

Iteration 0: Log likelihood = -9628.1621

Iteration 1: Log likelihood = -9628.1621

Computing standard errors ...

Mixed-effects ML regression

Group variable: country

Number of obs = 3,000

Number of groups = 30

Obs per group:

min = 100

avg = 100.0

max = 100

Wald chi2(3) = 370.90

Prob > chi2 = 0.0000

Log likelihood = -9628.1621

outcome	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
warmth	.8330937	.0574809	14.49	0.000	.7204332	.9457543
physical_punishment	-.9937819	.0798493	-12.45	0.000	-1.150284	-.8372801
2.intervention	.6406044	.2175496	2.94	0.003	.2142151	1.066994
_cons	52.65238	.4664841	112.87	0.000	51.73809	53.56668

Random-effects parameters	Estimate	Std. err.	[95% conf. interval]	
---------------------------	----------	-----------	----------------------	--

country: Identity					
	var(_cons)		3.371762	.9613269	1.928279 5.895816
-----+-----					
	var(Residual)		35.0675	.910002	33.32853 36.89721
-----					

LR test vs. linear model:  $\chi^2(01) = 204.14$       Prob  $\geq \chi^2 = 0.0000$

### 12.1.2 Generate Predicted Values

```
predict outcome_hat, fitted // predict yhat (`fitted` uses fixed AND random effects)
```

### 12.1.3 Graph With twoway Syntax

```
twoway (scatter outcome_hat warmth, mcolor(%30)) (lfit outcome_hat warmth)
graph export mypredictedvalues.png, width(1500) replace
twoway (lfit outcome_hat warmth)
graph export mypredictedvalues2.png, width(1500) replace
```

file mypredictedvalues.png saved as PNG format

file mypredictedvalues2.png saved as PNG format




Figure 8: Predicted Values From predict




Figure 9: Predicted Values From predict With Only Linear Fit

### 12.1.4 Spaghetti Plot With Predicted Values

```
spagplot outcome_hat warmth, id(country)
graph export myspaghetti2.png, width(1500) replace
```

file myspaghetti2.png saved as PNG format




Figure 10: Spaghetti Plot With Predicted Values

## 12.2 margins and marginsplot

In contrast to `predict`, which generates a predicted value for *every observation in the data*, `margins` generates predicted values at *specific values of certain variables*.

### 12.2.1 Estimate The Model

```
mixed outcome warmth physical_punishment i.intervention || country: // estimate MLM
```

Performing EM optimization Performing gradient-based optimization:  
 Iteration 0: Log likelihood = -9628.1621  
 Iteration 1: Log likelihood = -9628.1621

Computing standard errors ...

Mixed-effects ML regression  
 Group variable: country

Number of obs = 3,000  
 Number of groups = 30  
 Obs per group:  
     min = 100  
     avg = 100.0  
     max = 100

Wald chi2(3) = 370.90  
 Prob > chi2 = 0.0000

Log likelihood = -9628.1621

outcome	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
warmth	.8330937	.0574809	14.49	0.000	.7204332	.9457543
physical_p~t	-.9937819	.0798493	-12.45	0.000	-1.150284	-.8372801
2.interven~n	.6406044	.2175496	2.94	0.003	.2142151	1.066994
_cons	52.65238	.4664841	112.87	0.000	51.73809	53.56668

Random-effects parameters	Estimate	Std. err.	[95% conf. interval]	
country: Identity				
var(_cons)	3.371762	.9613269	1.928279	5.895816
var(Residual)	35.0675	.910002	33.32853	36.89721

LR test vs. linear model: chibar2(01) = 204.14      Prob >= chibar2 = 0.0000

### 12.2.2 Generate Predicted Values At Specified Values With margins

```
margins intervention, at(warmth = (1 2 3 4 5 6 7)) // predictive *margins*
```

Predictive margins      Number of obs = 3,000

Expression: Linear prediction, fixed portion, predict()

1.\_at: warmth = 1  
2.\_at: warmth = 2  
3.\_at: warmth = 3  
4.\_at: warmth = 4  
5.\_at: warmth = 5  
6.\_at: warmth = 6  
7.\_at: warmth = 7

		Delta-method					
		Margin	std. err.	z	P> z	[95% conf. interval]	
_at# intervention							
1	1	51.02222	.3966755	128.62	0.000	50.24475	51.79969
1	2	51.66283	.3955286	130.62	0.000	50.88761	52.43805
2	1	51.85532	.3788571	136.87	0.000	51.11277	52.59786
2	2	52.49592	.3789096	138.54	0.000	51.75327	53.23857
3	1	52.68841	.3692182	142.70	0.000	51.96476	53.41207
3	2	53.32902	.370554	143.92	0.000	52.60274	54.05529
4	1	53.52151	.3684014	145.28	0.000	52.79945	54.24356
4	2	54.16211	.3710204	145.98	0.000	53.43492	54.8893
5	1	54.3546	.376464	144.38	0.000	53.61674	55.09246
5	2	54.9952	.3802764	144.62	0.000	54.24988	55.74053
6	1	55.18769	.3928599	140.48	0.000	54.4177	55.95768
6	2	55.8283	.3977088	140.37	0.000	55.0488	56.60779
7	1	56.02079	.4166062	134.47	0.000	55.20425	56.83732
7	2	56.66139	.4223062	134.17	0.000	55.83369	57.4891

### 12.2.3 Graph With marginsplot

```
marginsplot // plot of predicted values  
graph export mymarginsplot.png, width(1500) replace
```

Variables that uniquely identify margins: warmth intervention

file mymarginsplot.png saved as PNG format






Figure 11: Predicted Values From margins and marginsplot

## 13 Scatterplot With Linear Fit and Marginal Density Plots (twoway ...)

As another possibility, we may wish to show more of the variation, by showing the variation in the *independent* variable and the *dependent* variable along with a *scatterplot* and *linear fit*. This is a complex graph and requires a little bit of manual programming in Stata.

You could also investigate the user written program `binscatterhist` (`ssc install binscatterhist`) which produces a similar looking graph, and automates much of this work.

### 13.1 Manually Generate The Densities To Plot Them Below (`kdensity ...`)

We generate the density for *warmth* at only a few points (`n(8)`) since this variable has relatively few categories.

```
kdensity warmth, generate(warmth_x warmth_d) n(8) // manually generate outcome densities
kdensity outcome, generate(outcome_y outcome_d) // manually generate outcome densities
```

## 13.2 Rescale The Densities So They Plot Well

You may have to experiment with the scaling and moving factors.

```
replace warmth_d = 100 * warmth_d // rescale the density so it plots well
replace outcome_d = 5 * outcome_d - .5 // rescale AND MOVE the density so it plots well
label variable outcome_y "density: beneficial outcome" // relabel y variable
```

(8 real changes made)

(50 real changes made)

## 13.3 Make The Graph (twoway ...)

You may have to experiment with whether scatterplots or line plots work best for displaying the x and y densities.

```
twoway (scatter outcome warmth, mcolor(%10)) /// scatterplot w some transparency
(lfit outcome warmth) /// linear fit
(line warmth_d warmth_x) /// line plot of x density
(line outcome_y outcome_d), /// line plot of y density (note flipped order)
title("Outcome by Warmth") /// title
ytitle("beneficial outcome") /// manual ytitle
xtitle("parental warmth") /// manual xtitle
legend(position(6) rows(2) ) /// legend at bottom; 2 rows
xlabel(0 1 2 3 4 5 6 7) /// manual x labels
name(mynewscatter, replace)

graph export mynewscatter.png, width(1500) replace
```

file mynewscatter.png saved as PNG format




Figure 12: Scatterplot and Linear Fit With Marginal Density Plots

### 13.4 Spaghetti Plot With Linear Fit and Marginal Density Plots

## 14 Curvilinear and Linear Fits

### Random Effects

Hemken, Doug. 2023. *Statamarkdown: 'Stata' Markdown*. <https://CRAN.R-project.org/package=Statamarkdown>.